
De�nici�on de plantillas

CLIPS> (deftemplate alumno "Datos de alumnos"

(slot nombre

(type STRING)

(default ?DERIVE))

(slot nota

(allowed-symbols suspenso aprobado notable

sobresaliente))

(slot repetidor

(allowed-symbols si no)

(default no)))

CLIPS> (list-deftemplates)

initial-fact

alumno

For a total of 2 deftemplates.

CLIPS> (assert (alumno))

<Fact-0>

CLIPS> (assert (alumno (nombre "Javier Romero")))

<Fact-1>

CLIPS> (facts)

f-0 (alumno (nombre "")

(nota suspenso)

(repetidor no))

f-1 (alumno (nombre "Javier Romero")

(nota suspenso)

(repetidor no))

For a total of 2 facts.

Adici�on de hechos con plantillas

CLIPS> (defrule escribe-notas

(alumno (nombre ?nombre)

(nota ?nota))

=>

(printout t "Nombre: " ?nombre crlf

"Nota: " ?nota crlf))

CLIPS> (assert (alumno (nombre "Javier Romero") (nota aprobado)))

<Fact-0>

CLIPS> (run)

Nombre: Javier Romero

Nota: aprobado

Eliminaci�on de hechos con plantillas

CLIPS> (undefrule *)

CLIPS> (defrule elimina-suspensos

?alumno <- (alumno (nombre ?nombre)

(nota suspenso))

=>

(retract ?alumno)

(printout t "Alumno suspendido: "

?nombre crlf))

CLIPS> (assert (alumno (nombre "Fernando Martinez")

(nota suspenso))

(alumno (nombre "Martin Fernandez")

(nota sobresaliente)))

<Fact-2>

CLIPS> (facts)

f-0 (alumno (nombre "Javier Romero")

(nota aprobado) (repetidor no))

f-1 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor no))

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

For a total of 3 facts.

CLIPS> (run)

Alumno suspendido: Fernando Martinez

CLIPS> (facts)

f-0 (alumno (nombre "Javier Romero")

(nota aprobado) (repetidor no))

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

For a total of 2 fact.

CLIPS>

Modi�caci�on de hechos con plantillas

CLIPS> (reset)

CLIPS> (undefrule *)

CLIPS> (defrule repetir

?alumno <- (alumno (nombre ?nombre)

(nota suspenso)

(repetidor no))

=>

(modify ?alumno (repetidor si)))

CLIPS> (assert (alumno (nombre "Fernando Martinez")

(nota suspenso))

(alumno (nombre "Martin Fernandez")

(nota sobresaliente)))

<Fact-2>

CLIPS> (facts)

f-0 (initial-fact)

f-1 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor no))

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

For a total of 3 facts.

CLIPS> (run)

CLIPS> (facts)

f-0 (initial-fact)

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

f-3 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor si))

For a total of 3 facts.

Duplicaci�on de hechos con plantillas

CLIPS> (reset)

CLIPS> (undefrule *)

CLIPS> (defrule repetir

?alumno <- (alumno (nombre ?nombre)

(nota suspenso)

(repetidor no))

=>

(duplicate ?alumno (repetidor si)))

CLIPS> (assert (alumno (nombre "Fernando Martinez")

(nota suspenso))

(alumno (nombre "Martin Fernandez")

(nota sobresaliente)))

<Fact-2>

CLIPS> (facts)

f-0 (initial-fact)

f-1 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor no))

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

For a total of 3 facts.

CLIPS> (run)

CLIPS> (facts)

f-0 (initial-fact)

f-1 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor no))

f-2 (alumno (nombre "Martin Fernandez")

(nota sobresaliente) (repetidor no))

f-3 (alumno (nombre "Fernando Martinez")

(nota suspenso) (repetidor si))

For a total of 4 facts.

Plantilla con rango

CLIPS> (clear)

CLIPS> (deftemplate alumno "Datos de alumnos"

(multislot nombre

(type SYMBOL)

(default ?NONE))

(slot nota

(type NUMBER)

(range 0 10)))

CLIPS> (assert (alumno))

[TMPLTRHS1] Slot nombre requires a value because of its

(default ?NONE) attribute.

CLIPS> (assert (alumno (nombre Ana Diaz)))

<Fact-0>

CLIPS> (facts)

f-0 (alumno (nombre Ana Diaz) (nota 0))

For a total of 1 fact.

CLIPS> (assert (alumno (nombre Luis Lara) (nota 11)))

[CSTRNCHK1] A literal slot value found in the

assert command does not fall in the allowed

range 0 to 10 for slot nota.

CLIPS> (assert (alumno (nombre "Luis Lara") (nota 1)))

[CSTRNCHK1] A literal slot value found in the

assert command does not match the allowed

types for slot nombre.

CLIPS>

Sintaxis de plantillas (I)

(deftemplate <nombre> [<comentario>]

<definicion-de-ranura>*)

<definicion-de-ranura>

::= <definicion-de-ranura-simple> |

<definicion-de-ranura-multiple>

<definicion-de-ranura-simple>

::= (slot <nombre-de-ranura>

<atributo-de-ranura>*)

<definicion-de-ranura-multiple>

::= (multislot <nombre-de-ranura>

<atributo-de-ranura>*)

<atributo-de-ranura>

::= <atributo-defecto> |

<restriccion-atributo>

<atributo-defecto>

::= (default ?DERIVE | ?NONE)

<restriccion-atributo>

::= (type <tipo>) |

<atributos-permitidos> |

<rango-de-atributo>

<tipo>

::= SYMBOL | STRING | NUMBER |

INTEGER | FLOAT

Sintaxis de plantillas (II)

<atributos-permitidos>

::= (allowed-symbols <simbolo>+) |

(allowed-strings <cadena>+) |

(allowed-numbers <numero>+) |

(allowed-integers <entero>+) |

(allowed-floats <flotante>+) |

(allowed-values <constante>+)

<rango-de-atributo>

::= (range <inicial> <final>)

<inicial> ::= <numero>

<final> ::= <numero>

